
Abstract

Notes for the workshop on “Plane quartics, Scorza map and related
topics”, held in Catania, January 19-21, 2016. The last section con-
tains eight Macaulay2 scripts on theta characteristics and the Scorza
map, with a tutorial. The first sections give an introduction to these
scripts.

1 Introduction

1.1 How to write down plane quartics and their theta
characteristics

Plane quartics make a relevant family of algebraic curves because their plane
embedding is the canonical embedding. As a byproduct, intrinsic and pro-
jective geometry are strictly connected. It is not a surprise that the theta
characteristics of a plane quartic, being the 64 square roots of the canonical
bundle, show up in many projective constructions. The first well known fact
is that the 28 odd theta characteristics correspond to the 28 bitangents of
the plane quartic curve. It is less known that the 36 even theta characteris-
tics may be visualized as the 36 plane quartic curves that are the preimages
through the Scorza map.

To write down explicitly a plane quartic and its theta characteristics, we
will adopt the following descriptions

1. • A homogeneous polynomial f of degree 4 in x0, x1, x2. Computation-
ally, this a vector with 15 homogeneous coordinates.

2. • A symmetric linear determinantal representation of f , namely a
symmetric 4 × 4 matrix A with linear entries in x0, x1, x2, such that
det(A) = f . The exact sequence

0−→O(−2)4
A−→O(−1)4−→θ−→0

gives a even theta characteristic θ on the curve f . For a general f ,
there are 36 classes of matrices A such that det(A) = f is given, up
to GL(4)-congruence, corresponding to the 36 even theta character-
istics. Computationally, we have a net of symmetric 4 × 4 matrices
〈A0, A1, A2〉.

3. • Let C = {f = 0}. A symmetric (3, 3)-correspondence in C × C has
the form Tθ = {(x, y) ∈ C × C|h0(θ + x − y) 6= 0} corresponding to a
even theta characteristic θ. Computationally, the divisor Tθ is cut on
the Segre variety P2 × P2 by six bilinear equations in x, y.

4. • Let {P1, . . . , P7} be seven general points in P2. The net of cubics
through these seven points defines a 2 : 1 covering P2(P1, . . . , P7)

π−→P2,
with source the blow-up of P2 at the points Pi, which ramifies over a
quartic C ⊂ P2. In equivalent way, seven general lines in P2 define a
unique quartic curve C such that the seven lines give a Aronhold system
of bitangents of C. Computationally, these data can be encoded in a
7 × 3 matrix (up to reorder the rows) or also in a 2 × 3 matrix of the
following type (

l0 l1 l2
q0 q1 q2

)
where deg li = 1, deg qi = 2 and li are independent. The maximal
minors of this matrix vanish on seven points.

Previous descriptions have an increasing amount of data, in the sense
that 2. and 3. are equivalent, while for the other items we have

{4.seven points} =⇒︸︷︷︸
8:1

{ 2. symmetric determinant
3. symm. (3,3) corresp.

} =⇒︸︷︷︸
36:1

{1.quartic polynomial}

(1.1)
By forgetting the theta characteristic in the intermediate step, we get the

interesting correspondence

{seven points} =⇒︸︷︷︸
288:1

{quartic polynomial} (1.2)

We will see how to “move on the right” in the diagrams (1.1) and (1.2),
from one description to another one. These moves, connecting the several
descriptions, are SL(3)-equivariant. Any family of quartics that is invariant
by the SL(3)-action of projective linear transformations can be described by
invariants or covariants in the above descriptions.

From the real point of view, it is interesting to recall the following Table
from [15, Prop. 5.1] for the theta characteristics of real quartics

topological classification # real odd theta # real even theta

empty 4 12
one oval 4 4

two nested ovals 4 12
two non nested oval 8 8

three ovals 16 16
four ovals 28 36

(1.3)

1.2 Clebsch and Lüroth quartics

The most famous example is given by Clebsch quartics, which are the quartics
which can be written as f =

∑5
i=1 l

4
i (Waring decomposition), while the

general quartic needs six summands consisting of 4th powers of linear forms,
in contrast with the naive numerical expectation. Clebsch quartics can be
detected by a determinantal invariant of degree 6 in the 15 coefficients of the
quartic polynomial (description 1.), called the catalecticant invariant or the
Clebsch invariant.

Description of special quartics may be quite different depending on the
different descriptions we choose. The archetypal example is that of Lüroth
quartics, which by definition contain the 10 vertices of a complete pentalat-
eral, like in the picture

The pentalateral defines in a natural way a particular even theta characteris-
tic θ, that it is called the pentalateral theta (the 10 vertices of the pentalateral
make the divisor 2K + θ). The pair (f, θ) consisting of a Lüroth quartic f
with a pentalateral theta θ can be described relatively easy looking at the de-
terminantal representation (description 2.) or even with the additional data

of a Aronhold system of seven bitangents (description 4.). The seven points
which give a Lüroth quartic correspond sursprisingly to the seven eigenvec-
tors of a plane cubic. If we forget these additional data, it is tremendously
difficult to detect a Lüroth quartic looking just at it defining polynomial
(description 1.).

Lüroth quartic were studied deeply in the period 1860-1918, starting from
Lüroth paper and their description in terms of net of quadrics, culminating
with Morley brilliant description of the invariant in terms of the seven points,
showing finally that the degree of Lüroth invariant, in terms of the fifteen
coefficients of the quartic, is 54. Lüroth quartics became again popular in
1977 when Barth showed that the jumping curve of a stable 2-bundle on
P2 with Chern classes (c1, c2) = (0, 4) is a Lüroth quartic. LePotier and
Tikhomirov showed in 2001 that the moduli space of the above bundles can
be described in terms of Lüroth quartics, the degree 54 turned out to be a
Donaldson invariant of P2.

1.3 The Scorza map

The Scorza map associates to a general quartic f a pair (S(f), θ) where S(f)
is another quartic (the Aronhold covariant) and θ is a even theta character-
istic on S(f). Its precise definition needs the Aronhold invariant of plane
cubics and it will be recalled in sectiobn 5. Its main property is the Theorem
of Scorza that

f 7→ (S(f), θ)

is dominant on the variety of pairs (g, θ) where θ is a even theta characteristic
on the quartic g. In a previous paper, Scorza showed that if f is Clebsch
then S(f) is Lüroth with pentalateral θ. This fact is the first step in the
proof of the Theorem of Scorza. We will give a computational description of
the Scorza map and its inverse, especially in Algorithms 6 and 7.

1.4 Description of the content

Sections from 2 to 6 describe a few basics about theta characteristics and
Scorza map for plane quartics. The goal is to introduce the terminology to
understand the algorithms, we refer to the literature for most of the proofs.
Some emphasis is given to Clebsch and Lüroth quartics, which correspond
through the Scorza map. More emphasis is given on the construction of a

quartic from seven points, which give a Aronhold system of seven bitangents.
Section 7 is a brief survey about invariant theory of plane quartics. Section 8
considers the link with the seven eigenvectors of a plane cubic. The reason we
include this section is that seven points are eigenvectors of a plane cubic with
respect to some nondegenerate quadric if and only if they define a Lüroth
quartic, according to (1.2). Section 9 is the core of these notes and contains
the Macaulay2[14] scripts.

1.5 Summary of the eight M2 scripts presented in §9

1. INPUT: seven general lines l1, . . . , l7

OUTPUT: the quartic having l1, . . . , l7 as Aronhold system of bitan-
gents

2. INPUT: a 2× 3 matrix with 2-minors vanishing on Z = {l1, . . . , l7}
OUTPUT: the quartic having Z as Aronhold system of bitangents

3. INPUT: seven general lines l1, . . . , l7

OUTPUT: a 8× 8 symmetric matrix (the bitangent matrix) collecting
in each row the 8 Aronhold systems of bitangents for the quartic having
l1, . . . , l7 as Aronhold system of bitangents, equivalent to l1, . . . , l7. In
particular, every principal 4× 4 minor of the bitangent matrix gives a
symmetric determinantal representation.

4. INPUT: a 2× 3 matrix with 2-minors vanishing on Z = {l1, . . . , l7}
OUTPUT: a symmetric determinantal 4×4 representation of the quar-
tic having Z as Aronhold system of bitangents

5. INPUT: a quartic f

OUTPUT: the image S(f) through the Scorza map

6. INPUT: a quartic f and a point q ∈ S(f)

OUTPUT: a determinantal representation corresponding to the image
(S(f), θ) through the Scorza map

7. INPUT: a determinantal representation of a quartic g corresponding to
(g, θ)

OUTPUT: the quartic f such that the image (S(f), θ) through the
Scorza map corresponds to (g, θ)

The tutorial contains a list of the 36 Scorza preimages of the Edge
quartic

8. INPUT: a plane quartic f

OUTPUT: the order of the automorphism group of linear invertible
transformations which leave f invariant

The algorithms 4, 6, 7 are computationally expensive. I wonder if there are
simpler solutions and shortcuts, from the computational point of view.

1.6 Acknowledgements

These notes were prepared for the Workshop on Plane quartics, Scorza map
and related topics, held in Catania, January 19-21, 2016. I warmly thank
Francesco Russo for the idea and the choice of the topic and all partici-
pants for the stimulating atmosphere. Special thanks to Edoardo Sernesi
and Francesco Zucconi for their very nice lectures [32, 35] who gave the
theoretical framework and allowed me to concentrate on the computational
aspects. Algorithm 4 was presented as an open problem in Catania, the
idea for its solution, with the selection of two cubics and the two additional
points were they vanish, is due to Edoardo Sernesi. I am deeply indebted to
Edoardo and his insight for my understanding of plane quartics. Algorithm
7 arises from a question discussed with Bernd Sturmfels. These notes owe
a lot to the computational point of view of the paper [28] by D. Plaumann,
B. Sturmfels and C. Vinzant. The topological classification of the 36 Scorza
preimages of the Edge quartic (see Algorithm 7) was computed by Emanuele
Ventura, after the workshop.

2 Apolarity, Waring decompositions

Our base ring is S∗(V ) = C[x0, x1, x2]. The dual ring of differential operators
is S∗(V ∨) = C[∂0, ∂1, ∂2] with the action satisfying ∂i(xj) = δij.

A differential operator g ∈ S∗(V ∨) such that g · f = 0 is called apolar to
f . Differential operators of degree d can be identified with plane curves of
degree d in the dual plane.

Apolarity is very well implemented in M2 by the command diff, with the
caveat that differential operators are written with the same variables xi of
the ring where they act.

Definition 2.1 Denote Pa =
∑

i ai∂i. The polar of f ∈ SymdV at a is
Pa(f) ∈ SymdV .

If f corresponds to the symmetric multilinear form f(x, . . . , x) then Pa(f)
corresponds to the multilinear form f(a, x, . . . , x). It follows that after d
iterations we get

P d
a (f) = d!f(a).

An example important for Morley construction is the following, which is
discussed in [26]. Take a cubic f with a nodal point Q. Then Pa(f) is the
nodal conic consisting of the two nodal lines making the tangent cone at Q.

Note that f depends essentially on ≤ 2 variables (namely it is a cone) if
and only if there is a differential operator of degree 1 apolar to f , in this case
we say that there is a line apolar to f .

The lines apolar to f make the kernel of the contraction map

C1
f : V ∨ → Sym3V

In the same way, the conics apolar to f make the kernel of the contraction
map

C2
f : Sym2V ∨ → Sym2V

The map C2
f is called the middle catalecticant map and there are conics

apolar to f if and only if the middle catalecticant of f vanishes.
In equivalent way, the Clebsch quartics f of section 1.2 can be defined by

the condition detC2
f = 0.

3 The Aronhold invariant of plane cubics

References for this section: [23, 34, 9, 17].
The Aronhold invariant is the equation of the SL(3)-orbit of the Fermat

cubic x3 + y3 + z3 in P9 = P(Sym3C3). By construction, it is an SL(3)-
invariant in the 10 coefficients of a plane cubic. In other words, it is the
equation of the 3-secant variety to the 3-Veronese embedding of P2, which is
an hypersurface of degree 4.

Theorem 3.1 Let f(x, y, z) ∈ Sym3C3 be a homogeneous cubic polynomial.
Let C(fx), C(fy), C(fz) be the three 3× 3 symmetric matrices which are the
Hessian of the three partial derivatives of f .

All the 8-pfaffians of the 9× 9 skew-symmetric matrix 0 C(fz) −C(fy)
−C(fz) 0 C(fx)
C(fy) −C(fx) 0

 (3.1)

coincide (up to scalar) with the Aronhold invariant.

Let End0 C3 be the space of traceless endomorphisms of C3. The matrix
(3.1) describes the contraction

Af : End0 C3 → End0 C3,

which in the case f = v3 satisfies

Av3(M)(w) = (M(v) ∧ v ∧ w) v ∀M ∈ End C3, ∀w ∈ C3.

Theorem 3.2 (Nonabelian apolarity for plane cubics) Let f = l31 +
l32 + l33. Then li are (symultaneous) eigenvectors of all the matrices M ∈
kerAf ⊂ End0 C3.

The explicit expression of the Aronhold invariant Ar (sometimes called also
S) has 25 monomials and it can be found in [34] Prop. 4.4.7 or in [10] (5.13.1),
or as output of the following M2 script

R=QQ[x,y,z,c_0..c_9]

x1=matrix{{x,y,z}}

x3=symmetricPower(3,x1)

f=(matrix{{c_0..c_9}}*transpose x3)_(0,0)

m=matrix{{0,z,-y},{-z,0,x},{y,-x,0}}

---following is 9*9 matrix

m9=diff(m,diff(x1,diff(transpose x1,f)))

---following is Aronhold invariant

aronhold=(mingens pfaffians(8,m9))_(0,0)

By a slight abuse of notation we denote by Ar also the corresponding
multilinear form. The classical symbolic expression for Ar is

Ar(x3, y3, z3, w3) = (x ∧ y ∧ z)(x ∧ y ∧ w)(x ∧ z ∧ w)(y ∧ z ∧ w).

Proposition 3.3 The closure of the orbit SL(3) ·(x3 + y3 + z3) contains the
following three orbits

equation dim rk

Fermat x3 + y3 + z3 8 3
cuspidal y2z − x3 7 4

smooth conic+tg line x(y2 − xz) 6 5

These are all the cubics with border rank three.
The Aronhold invariant vanishes on the above three orbits and on all the

cubics depending on essentially one or two variables, they have border rank
≤ 2 and they are cones with a point as vertex.

In the classical terminology, a Fermat cubic l31+l32+l33 has a polar 3-lateral
given by l1l2l3. A recipe to compute the polar 3-lateral in the three cases of
Prop. 3.3 is given by the following Proposition.

Proposition 3.4 (i) The Hessian of l31 + l32 + l33 factors as l1l2l3.
(ii) The Hessian of a cuspidal cubic splits as l21l2 where l21 is the tangent

cone at the cusp.
(iii) The Hessian of {smooth conic} ∪ {tg line l} is l3.

Proposition 3.5 Real Fermat cubics[2] The real cubics with complex border
rank three make four SL(3,R)-orbits, the Fermat case in Prop. 3.3 splits
into the two cases

(i’) orbit of x3 + y3 + z3, real Fermat
(i”) orbit if (x+

√
−1y)3+(x−

√
−1y)3+z3 = 2x3−6x2y+z3, imaginary

Fermat

4 Three descriptions of an even theta char-

acteristic

4.1 The symmetric determinantal description

Let A0, A1, A2 be three symmetric 4× 4 matrices. The expression

det (xA0 + yA1 + zA2)

defines a plane quartic with a even theta θ given by

0−→OP2(−2)4
A−→OP2(−1)4−→θ−→0

As an example, the Edge quartic

f = 25x4 − 34x2y2 + 25y4 − 34x2z2 − 34y2z2 + 25z4 (4.1)

has the symmetric determinantal representation
0 x+ 2y 2x+ z y − 2z

x+ 2y 0 y + 2z −2x+ z
2x+ z y + 2z 0 x− 2y
y − 2z −2x+ z x− 2y 0


We refer to [32, 28] for the correspondence between the 28 bitangents of

the quartic and the lines joining the eight base points of the net of quadrics.
The seven lines through a fixed point make a Aronhold system of bitangents.
The seven corresponding points in P3 are Gale dual (see [28, 13]) of the seven
bitangents.

4.2 The sextic model

Let K be the canonical bundle over a smooth plane quartic. The line bundle
K + θ has 4 independent sections which give a linear system of effective
divisors of degree 6. It gives an embedding of the curve as a degree 6 (and
genus 3) curve in P3 = P (H0(K + θ)∨). It has the resolution

0−→OP3(−3)3
M−→OP3(−2)4

g−→OP3(1)−→K + θ−→0 (4.2)

The symmetric determinantal description in 4.1 gives a 4×4×3-tensor which
has another flattening in the 4 × 3 matrix M with four linear entries. The

four maximal minors of M give the cubic equations of the sextic and define
the map g in (4.2). The Edge quartic (9.1) gives the following

M =


u1 + 2u2 2u1 + u3 u2 − 2u3
u0 − 2u3 2u0 + u2 2u2 + u3
2u0 + u3 u1 − 2u3 u0 + 2u1
−2u1 + u2 u0 − 2u2 −2u0 + u1


The isomorhism between the quartic model with coordinates (x, y, z) and

the sextic model with coordinates (u0, . . . , u3) is guaranteed by the system

M(u) · (x, y, z)t = 0. (4.3)

Given u such that rkM(u) = 2, the system (4.3) defines a unique (x, y, z).
Conversely, given (x, y, z) on the quartic model, the system (4.3) defines a
unique (u0, . . . , u3).

The M2 code to construct M from A is the following

R=QQ[x,y,z,u_0..u_3]

A= matrix({{0, x + 2*y, z + 2*x, -2*z + y},

{x + 2*y, 0, 2*z + y, z - 2*x},

{z + 2*x, 2*z + y, 0, x - 2*y},

{-2*z + y, z - 2*x, x - 2*y, 0}});

f=det(A)

uu=transpose matrix {{u_0..u_3}}

M=diff(x,A)*uu|diff(y,A)*uu|diff(z,A)*uu

An elegant and alternative way to construct the divisor K + θ is the
following

Theorem 4.1 (Dixon) Given a symmetric determinantal representation A,
the four principal minors define four cubics which are contact cubics, namely
they cut the quartic det(A) in a nonreduced divisor supported on a degree six
divisor which is K + θ.

More generally, if Aadj is the adjugate matrix, utAadju parametrizes con-
tact cubics, making a 3-fold of degree 8 in P9.

From the sextic model, one finds four independent sections ui ∈ H0(K +
θ). Then the sections uiuj may be lifted to cubics in P2, with proper scaling

guaranteed by the equation (ui + uj)
2 = u2i + u2j + 2uiuj. Then the adjugate

of the 4×4 matrix (uiuj) has degree 9 polynomials which contain f 2 as a fac-
tor, after dividing by f 2 we get the corresponding symmetric determinantal
description.

Remark 4.2 The Dixon’s Theorem 4.1 generalizes to d× d symmetric rep-
resentation, in this case the degree d(d− 1)/2 divisor which is the support of
the nonreduced divisor is H + θ.

Remark 4.3 Note that a general 4 × 3 matrix M with linear entries in
u0, . . . , u3 defines a sextic curve of genus 3, but the line bundle which gives
the embedding has the form K+L, with L a degree 2 line bundle which is not
necessarily a theta characteristic. The condition to be a theta characteristic is
equivalent to the fact that the 4×4 matrix with entries in x0, . . . , x2 obtained
by flattening M may be symmetrized by row/columns operations.

4.3 The (3, 3)-correspondence

References for this section: [8, 10]. Let θ be an even theta characteristic on
C of genus 3. It is defined a (3, 3) correspondence from the following divisor
on C × C

Tθ = {(P,Q) ∈ C × C|h0(θ + P −Q) > 0}. (4.4)

It follows from Serre duality that the correspondence is symmetric. By
Riemann-Roch, ∀P ∈ C, θ + P is linearly equivalent to a unique effective
divisor of degree 3, hence we get a (3, 3) correspondence.

Remark 4.4 An analogous (g, g) correspondence may be defined starting
from a general line bundle L of degree g − 1, but this correspondence is
in general not symmetric, the symmetry is guaranteed from L being a theta
characteristic (see Remark 4.3).

In the sextic model the correspondence has the following form: pick co-
ordinates (u0, . . . , u3), (v0, . . . , v3) and we have, for any (u, v) ∈ C × C

(u, v) ∈ Tθ ⇐⇒ v ·M(u) = 0.

This works because kerM(u) is a 3-secant line of the sextic model (see
[31]).

5 The Aronhold covariant of plane quartics

and the Scorza map.

Main references for this section are [10, 22, 26]. Recall by Definition 2.1 that
Pxf is the polar of f at the point x. The Scorza map is defined as f 7→
(S(f), θ), where S(f) = {x ∈ P2|Ar(Pxf) = 0} is the Aronhold covariant
(the notation with the letter S goes back to Clebsch and has nothing to do
with Scorza, see §7 in Ciani’s monograph Cia) and the (3, 3)-correspondence
Tθ which encodes θ as in (4.4) is defined by

{(x, y) ∈ P2 × P2|rk(PxPyf) ≤ 1}.

The main result regarding the Scorza map is the following Theorem,
proved by Scorza in 1899[31].

Theorem 5.1 (Scorza) The map f 7→ S(f) is dominant and is generically
36 : 1.

Since there are 36 even theta characteristic on a general quartic curve,
Theorem 5.1 implies that the general pair (C, θ) where C is a plane quartic
and θ is an even theta characteristic on C comes from a unique quartic f
through the Scorza map.

Assume that Pxf = l31 + l32 + l33 and denote xij = {li = lj = 0}. The
divisor θ is linearly equivalent to the divisor

x12 + x13 + x23 − x.

This description allows to compute explicitly both the Scorza map and
its inverse (see the Algorithm 7). Note that the determinantal description
can be obtained in the coefficient field of f if S(f) contains a point x lying
in the same field.

Theorem 5.2 If f is Clebsch then (S(f), θ) is Lüroth with the pentalateral
θ. Conversely the general (g, θ), where g is a Lüroth curve with pentalateral
θ, comes from a unique Clebsch curve f such that S(f) = g.

Corollary 5.3 The infinitely many decomposition of a Clebsch quartic, with
f =

∑4
i=0 l

4
i and li circumscribed to the conic C, give infinitely many penta-

lateral inscribed in S(f).

A reflection on this Theorem allows to understand why the invariant
description of Lüroth condition for the pair (f, θ) (regarding net of quadrics)
is much simpler than the one for f itself.

Since the Scorza map is SL(3)-equivariant, we remark the following con-
sequence

Proposition 5.4 Let Aut(f) be the automorphsim group of linear transfor-
mation of P2 leaving the quartic f invariant. Then

Aut(f) ⊆ Aut(S(f))

is a group inclusion. In particular the order of Aut(f) divides the order of
Aut(S(f)).

As a corollary, note that both double conics (having Aut(f) = SL(2))
and Klein quartic (having Aut(f) the simple group of order 168, the group of
higher order among all irreducible quartics) both satisfyAut(f) = Aut(S(f)).

Moreover, the 36-preimages S−1S(f)divide into SL(3)-orbits Oi for i =
1, . . . , k, which are the same as Aut(S(f)). These orbits are studied in the
literature regarding the action of the automorphism group on the even spin
structures of the curve.

For any f ∈ Oi the size of Aut(f) is fixed and we get

|Oi| =
|Aut(S(f))|
|Aut(f)|

,

which can be used jointly with the obvious identity

k∑
i=1

|Oi| = 36.

See the comments and the tables before the M2 script of Algorithm 7 in
§9.

6 Contact cubics and contact triangles

For any even theta characteristics θ, the effective divisors of degree six cor-
responding to K + θ can be computed with the following trick. Let A be
a 4 × 4 symmetric determinantal representation corresponding to θ, as in

4.1. Any principal minor of A defines a contact cubic, which cuts the quar-
tic in a nonreduced divisor, supported on K + θ. More generally, for any
u = (u0, . . . , u3) there is a contact cubic given by

det

(
A ut

u 0

)
. (6.1)

This was the classical formula which gives the entries of the adjugate ma-
trix of A, it is quite convenient from the computational point of view when
regarding matrices with symbolic entries.

When diagonal elements of A are zero, we have a further description,
contact cubics are triangles given by three bitangents, like in

0 l01 l02 l03
l01 0 l12 l13
l02 l12 0 l23
l03 l13 l23 0


where lij are bitangents. In this case the four principal minors are contact
triangles, like  0 l01 l02

l01 0 l12
l02 l12 0


which gives l01l02l12.

There are 56 contact cubics given by three bitangents in the family (6.1)
of contact cubics, all together they are 56·36 = 2016. All of them have the six
contact points which do not lie on a conics. The 36 families of contact cubics
correspond to 36 2-Veronese 3folds in P9 (of degree 8). They do not meet the
2-secant variety of the 3-Veronese surface, namely triangles giving by three
collinear lines. They meet the variety of triangles in 120 points. There are
8 strictly biscribed triangles, according to Mukai[21], each one counts with
multiplicity 8 for a total of 64, indeed note that 56+64 = 120 = 8 ·15 (degree
of intersection of Veronese 3fold with the variety of triangles).

There are other 28 families of contact cubics, given by the projection-
Veronese 3fold projected (from the bitangent) on a hyperplane. In each of
these familes there are 45 contact triangles given by three bitangents. Note
that 45 · 28 = 1260 and 1260 + 2016 = 3276 =

(
28
3

)
and we have counted

exactly once all triples of bitangents. All the 1260 triples have six contact
points on a conic. They meet the 3-Veronese surface of triple lines in one

point of multiplicity 6, which corresponds to the bitangent at power 3, like
l3. In any case, the expression ml3 + q2 = f does not hold for any line m and
conic q. Label the 28 bitangents as ij where 1 ≤ i < j ≤ 8, corresponding
to the pairs of base points of the net of quadrics, as in 4.1. These 45 contact
triangles divide in two types, 30 of them like 12.23.34 other 15 of them like
12.34.56 (see [5, §13]). In the first type there is a fourth bitangent 41 such
that the 8 contact points lie on a conic, there are 210 4ples of this kind, also
in the second type there is a fourth bitangent 78 such that the 8 contact
points lie on a conic, there are 105 4ples of this kind. Altogether, there are
210 + 105 = 315 4ples of bitangents such that the 8 contact points lie on a
conic.

7 The invariant ring of plane quartics

The complete determination of invariant ring of plane quartics is a big
achievement of computer algebra, the final step was presented at MEGA
2013 in Frankfurt by Andreas-Stephan Elsenhans [12], relying on previous
work by Shioda and Dixmier, so solving a classical question which went back
to Emmy Noether’s doctoral thesis.

The following result was conjectured by Shioda in 1967, who computed
the Hilbert series.

Theorem 7.1 The invariant ring
[
⊕dSymd

(
Sym4C3

)]SL(3)
is generated by

invariants of degree 3, 6, 9, 9, 12, 12, 15, 15, 18, 18, 21, 21, 27. The relations are
known.

Dixmier found in 1987 the invariants of degree 3, 6, 9, 12, 15, 18, 27 which are
algebraically independent, so that the invariant ring is an algebraic extension
of the ring generated by these ones (primary invariants). The invariants up
to degree 18 (more the discriminant of degree 27) can be found in Salmon
book[30], compare also with see [5, §7]. So only the invariants of degree
21 were missing in XIXth century. Apparently the first who produced the
complete generators of invariant ring was T. Ohno in an unpublished work in
2007. Note the elementary fact that the degree of any invariants is divisible
by 3. The Clebsch invariant, defining quartics of rank 5, has degree 6. Some
classical facts regarding the cubic invariant are recalled in [25].

8 The link with the seven eigentensors of a

plane cubic

Given a plane cubic f ∈ Sym3V ∨, an eigentensor of f is v ∈ V such that

f(v, v, x) = λq(v, x) (8.1)

for every x ∈ V . When q is the euclidean metric (in the real setting), which
gives an identification between V and V ∨, the previous equation can be
written as f(v2) = λv, which is the way the eigentensor equation is commonly
written in the numerical setting, and it is the natural generalization of the
eigenvector condition for symmetric matrices.

In the metric setting, the generalization is more transparent. Indeed, the
eigenvectors of a symmetric matrix q are the critical points of the distance
function from q to the Veronese variety (P2,O(2)). In the same way, the
seven eigentensors are the seven critical points of the distance function from
f to the Veronese variety (P2,O(3)) , see [11].

A dimensional count shows that seven eigenvectors of a cubic cannot
be seven general point, so that it is interesting to understand their special
position according to (1.2). The following is the geometric counterpart of [1,
Prop. 5.3].

Theorem 8.1 (Bateman) (i) The seven points p1, . . . , p7 which are eigen-
tensors of a cubic satisfy the following property:

the seven nodal conics Ci for i = 1, . . . 7 which correspond to PPiHi, where
Hi is the unique cubic passing through all pj and singular at pi are harmonic,
that is ∆Ci = 0, where ∆ is the Laplacian.

(ii) Seven points vi are eigentensors of a cubic f with respect to some
nondegenerate conic q as in (8.1) if and only the seven points give a Lüroth
quartic in the correspondence (1.2).

Proof. Part (i) is a reformulation of Morley differential identity, see
formula (20) in §9 of [26], where θ = x20 +x21 +x22 and the fact that (with the
notations of [26]), the Morley form M(Pi, X) coincides with Hi(X), see the
paragraph after Corollary 3.3 in [26]. Part (ii) is a reformulation of Theorem
10.4 in [26]

9 Eight algorithms and Macaulay2 scripts,

with a tutorial

Algorithm 1

• INPUT: seven general lines l1, . . . , l7

• OUTPUT: the quartic having l1, . . . , l7 as Aronhold system of bitan-
gents

The steps of the algorithm are the following

1. Compute the 2× 3 matrix(
a0(x) a1(x) a2(x)
q0(x) q1(x) q2(x)

)
with deg ai = 1, deg qi = 2 degenerating on Z = {l1, . . . , l7} seen as
seven points in the dual space. This matrix is computed from the
resolution of the ideal vanishing on the points and it can be taken as
input (as in Algorithm 2).

2. Construct the net of cubics passing through Z as

net(x, y) = det

a0(y) a1(y) a2(y)
a0(x) a1(x) a2(x)
q0(x) q1(x) q2(x)


3. Construct the jacobian of the net as the determinant of the 3×3 matrix

jac(x) = det

[
∂2net

∂xi∂yj

]
which is a sextic (in coordinates x) nodal at the seven points.

4. The quartic in output is obtained by eliminating x from the four equa-
tions jac(x) and yi − ∂net(x,y)

∂yi
.

As a running example, start with input given the seven points which are
the rows of the following matrix

1 2 0
2 0 1
0 1 −2
5 5 3
5 −3 5
3 5 −5
−1 1 1


get as output the quartic

f = 25y40 − 34y20y
2
1 + 25y41 − 34y20y

2
2 − 34y21y

2
2 + 25y42.

In alternative, start from(
x0 x1 x2

(3x20 + 2x0x1 + x22) (x20 + 2x1x2) (x21 + 2x0x2)

)
where the second row is the gradient of x20x1 + x21x2 + x22x0 + x30 and get

as output the Lüroth quartic

f = 2y30y1−2y20y
2
1−y0y31−y30y2−2y20y1y2+4y0y

2
1y2+2y31y2+y

2
0y

2
2−2y0y1y

2
2+y21y

2
2−y0y32−y1y32.

The M2 script is the following

KK=QQ

R=KK[x_0..x_2,y_0..y_2]

x1=matrix{{x_0..x_2}}

y1=matrix{{y_0..y_2}}

----the rows of p2 contain the seven points, below are shown two samples

p2=random(R^{7:0},R^{3:0})

p2=matrix{{1,2,0},{2,0,1},{0,1,-2},{5,5,3},{5,-3,5},{3,5,-5},{-1,1,1}}

---computation of the 2*3 matrix mat

I7=minors(2,p2^{0}||x1)

for i from 0 to 6 do I7=intersect(I7,minors(2,p2^{i}||x1))

r7=res I7

betti r7

mat=transpose r7.dd_2

------in alternative one can start from a 2*3 matrix as above

---we construct now the Morley form

nc=(sub((mat)^{0},apply(3,i->(x_i=>y_i)))||mat)

net2=diff(matrix{{y_0..y_2}},det(nc))

---jacobian of the net, is a plane sextic nodal at seven points

jac=ideal(det(diff(transpose x1,net2)))

-----get the quartic f eliminating x_i from the net of cubics and the jacobian

ff=eliminate({x_0,x_1,x_2},ideal(matrix{{y_0..y_2}}-net2)+jac)

f=((gens ff)_(0,0))

----f is our OUTPUT

---check the seven starting points are really bitangents

for i from 0 to 6 do print(i, degree ideal(f,(p2^{i}*transpose y1)_(0,0)),

degree radical ideal(f,(p2^{i}*transpose y1)_(0,0)))

Algorithm 2

• INPUT: a 2× 3 matrix with 2-minors vanishing on Z = {l1, . . . , l7}

• OUTPUT: the quartic having Z as Aronhold system of bitangents

This is just the combination of steps 2 . . . 4 of Algoritm 1.

A typical application is the matrix

(
x1 x2 x0
x20 x

2
1 x

2
2

)
which gives the Klein

quartic y0y
3
1 +y1y

3
2 +y30y2. Let τ = e2π

√
−1/7 a 7-th root of unity. In this case,

the seven lines in Z are represented by the columns of the matrix

(
y0 y1 y2

)
·

1 τ 3 τ 6 τ 2 τ 5 τ τ 4

1 τ 5 τ 3 τ τ 6 τ 4 τ 2

1 τ 2 τ 4 τ 6 τ τ 3 τ 5


but the script does not need τ and works on the field where the 2× 3 matrix
is defined.

The M2 script is the following

KK=QQ

R=KK[x_0..x_2,y_0..y_2]

x1=matrix{{x_0..x_2}}

y1=matrix{{y_0..y_2}}

-----with the following matrix we find Klein quartic !!

mat=matrix{{x_1,x_2,x_0},{x_0^2,x_1^2,x_2^2}}

nc=(sub((mat)^{0},apply(3,i->(x_i=>y_i)))||mat)

net2=diff(matrix{{y_0..y_2}},det(nc))

codim ideal net2, degree ideal net2

---jacobian of the net, is a plane sextic nodal at seven points

jac=ideal(det(diff(transpose x1,net2)))

-----get the quartic f eliminating x_i from the net of cubics and the jacobian

ff=eliminate({x_0,x_1,x_2},ideal(matrix{{y_0..y_2}}-net2)+jac)

f=((gens ff)_(0,0))

---f is our output

Algorithm 3

• INPUT: seven general lines l1, . . . , l7

• OUTPUT: a 8× 8 symmetric matrix (the bitangent matrix) collecting
in each row the 8 Aronhold systems of bitangents for the quartic having
l1, . . . , l7 as Aronhold system of bitangents, equivalent to l1, . . . , l7. In
particular, every principal 4× 4 minor of the bitangent matrix gives a
symmetric determinantal representation.

These are the steps

1. Write the coordinates of the seven points {l1, . . . , l7} as columnsof a
3× 7 matrix P and compute a Gale transform in P3 given by the seven
rows {m1, . . .m7} of a 7 × 4 matrix M such that PM = 0. These are
well defined modulo SL(4)-action.

2. Compute the net of quadrics through {m1, . . .m7}, spanned by the 4×4
symmetric matrices Q0, Q1, Q2.

3. Compute the symmetric determinantal representation of the quartic
curve as det(

∑
i xiQi).

4. Compute (with a saturation), the eighth base point m0 of the net and
correspondingly stack it as a first row over M so obtaining the 8 × 4
matrix M0. The theory guarantees that the 28 lines mimj for 0 ≤ i <
j ≤ 7 are the 28 bitangent of the quartic constructed in the previous
step, but we have to apply a linear projective transformation to

5. Compute the linear projective transformation g from the space of lines
in P3 through m0 to our dual P2 which takes the line m0mi to li for
i = 1, . . . , 7.

6. The bitangent matrix is M0(
∑

i g(x)iQi)M
t
0 (see [28, (3.4)]).

An example is given by the following seven lines in input



x0
x1
x2

x0 + x1 + x2
x0 + 2x1 + 3x2
2x0 + 3x1 + x2
3x0 + x1 + 2x2


The corresponding quartic is

81x4
0 + 198x3

0x1 − 41x2
0x

2
1 − 198x0x

3
1 + 81x4

1 − 198x3
0x2 − 1561x2

0x1x2 − 1561x0x
2
1x2 +

198x3
1x2 − 41x2

0x
2
2 − 1561x0x1x

2
2 − 41x2

1x
2
2 + 198x0x

3
2 − 198x1x

3
2 + 81x4

2

and the bitangent matrix in output has the following first three columns
(the whole matrix is too big to be printed, it can be found by running the
M2 script).



0 3x0 3x1
3x0 0 (777/143)x0 + (1239/143)x1 − (378/143)x2
3x1 (777/143)x0 + (1239/143)x1 − (378/143)x2 0
3x2 (1239/143)x0 − (378/143)x1 + (777/143)x2 −(378/143)x0 + (777/143)x1 + (1239/143)x2

11x0 + 11x1 + 11x2 (126/13)x0 + (287/13)x1 + (133/13)x2 (133/13)x0 + (126/13)x1 + (287/13)x2
−x0 − 2x1 − 3x2 −(756/143)x0 − (448/143)x1 − (525/143)x2 (119/143)x0 − (189/143)x1 − (1113/143)x2
−2x0 − 3x1 − x2 −(189/143)x0 − (1113/143)x1 + (119/143)x2 −(392/143)x0 − (84/143)x1 − (161/143)x2
−3x0 − x1 − 2x2 −(84/143)x0 − (161/143)x1 − (392/143)x2 −(525/143)x0 − (756/143)x1 − (448/143)x2



The M2 script is the following

restart

R=QQ[x_0..x_2,y_0..y_3]

x1=matrix{{x_0..x_2}}

y2=symmetricPower(2,matrix{{y_0..y_3}})

y1=matrix{{y_0..y_3}}

p2=matrix{{1,2,0},{2,0,1},{0,1,-2},{5,5,3},{5,-3,5},{3,5,-5},{-1,1,1}}

--- p2 contains seven points in P2

------p3 contains the seven points in P^3 which are the Gale transform

p3=gens kernel transpose p2

---p3 contains the seven points in P3 which are Gale dual, defined up to PGL(4)

qp3=symmetricPower(2,p3^{0})

for i from 1 to 6 do qp3=qp3||symmetricPower(2,p3^{i})

net3=y2*gens kernel qp3

---net3 is the net of quadrics in P3

---note that even if the net is defined up to PGL(4),

----the locus of singular quadrics is well defined in the net

----the important fact is that the net of quadrics is dual to the original P^2

for i from 0 to 2 do Q_i=diff(transpose y1,diff( y1,net3_(0,i)))

--for i from 0 to 2 do print Q_i

f=det(sum(3,i->x_i*diff(transpose y1,diff( y1,net3_(0,i)))))

---f is the plane quartic curve, living in projective plane of nets of quadrics,

---it is not yet in the right coordinate system

----now we use the eighth point to make a projectivity with starting plane

pp3=minors(2,p3^{0}||y1)

for i from 1 to 6 do pp3=intersect(pp3,minors(2,p3^{i}||y1))

codim pp3, degree pp3

---pp3 is the ideal of seven points in P^3, obtained by Gale tranform

eighth=saturate(ideal(net3),pp3)

cord8=sub(matrix{apply(4,i->(y_i%eighth))},y_3=>1)

---cord8 contains coordinates of 8th point.

cc=sub(net3,apply(4,i->(y_i=>p3_(0,i)+2*cord8_(0,i))))

for i from 1 to 6 do cc=cc||sub(net3,apply(4,j->(y_j=>p3_(i,j)+2*cord8_(0,j))))

----cc contains the coordinates of the seven lines through the 8th point,

---in the coordinates of the net

---now we look for projectivity between cc and p2, in the SAME order of points

R1=QQ[g_0..g_8]

gg=transpose genericMatrix(R1,3,3)

-----following are set of points we want to make in projection

-----we impose conditions to a unknown projective transformation gg

p21=sub(p2,R1),cc1=sub(cc,R1)

IG=minors(2,p21^{0}*gg||cc1^{0})

for i from 1 to 3 do IG=IG+minors(2,p21^{i}*gg||cc1^{i})

codim IG, degree IG

for i from 4 to 6 do IG=IG+minors(2,p21^{i}*gg||cc1^{i})

codim IG, degree IG

gc=sub(matrix{apply(9,i->(g_i)%IG)},g_8=>1)

gm=submatrix(gc,{0..2})||submatrix(gc,{3..5})||submatrix(gc,{6..8})

----gm is the projective transformation found

det gm

gi=inverse (sub(gm,QQ))

ff=sub(f,apply(3,i->(x_i=>(gi*transpose x1)_(i,0))))

----ff is the quartic in the starting(correct) coordinate system

---qq is the determinantal representation of the quartic

-- with given Aronhold system of bitangents

qq=sum(3,i->((gi*transpose x1)_(i,0))*Q_i)

factor(det(qq))

---we want to find now the bitangent matrix, according to Hesse

---np3 cotains the eight base points of the net

np3=cord8||p3

btm=np3*qq*(transpose np3)*(1/20)

---btm is the bitangent matrix

---check that all 28 lines are bitangents

for i from 0 to 6 do for j from i+1 to 7 do print(i,j,

degree radical ideal(ff,btm_(i,j)))

Algorithm 4

• INPUT: a 2× 3 matrix with 2-minors vanishing on Z = {l1, . . . , l7}

• OUTPUT: a symmetric determinantal 4×4 representation of the quar-
tic having Z as Aronhold system of bitangents

Starting with the the matrix

(
x1 x2 x0
x20 x

2
1 x

2
2

)
which gives the Klein quartic

y0y
3
1 + y1y

3
2 + y30y2 we find the following determinantal representation

2x2 0 0 −x1
0 2x0 0 −x2
0 0 2x1 −x0
−x1 −x2 −x0 0


These are the steps of the algorithm

1. Compute the net of cubics through the seven points and choose two
independent cubics in the net. In the matrix description, these are
given by two 2× 2 minors.

2. With a saturation, compute the two additional points where the two cu-
bics vanish and compute the 4-dimensional space of conics 〈Q0(x), . . . Q3(x)〉
vanishing at these two points.

3. A Gale tranform in P3 can be found by eliminating xi from the net
of cubics and from the four equations yi − Qi(x) for i = 0, . . . 3. The
result is an ideal p3 in coordinates y generated by three quadrics and
one cubic, vanishing in seven points. With these conventions, the net
net3 of the three quadrics 〈F0(y), . . . , F2(y)〉 vanish on a eighth point
which has coordinates (y0, y1, y2, y3) = (0, 0, 0, 1).

4. We eliminate yi from the ideal generated by xi − Fi(y) and from the
three cubics(in y) obtained by eliminating y3 from p3. The result is a
net of cubics (in xi) that we call I7net.

5. We find a linear 3× 3 transformation in coordinates xi from the net of
plane cubics we started with, to the net I7net, after this linear trans-
formation is applied to Fi(y), the resulting net of quadrics gives the
determinantal representation in output.

KK=QQ

R=KK[x_0..x_2]

x1=matrix{{x_0..x_2}}

mat=matrix{{x_1,x_2,x_0},{x_0^2,x_1^2,x_2^2}}

----we redefine I7 accordingly, so that I7 comes from mat

I7=gens minors(2,mat)

---chose two cubics in the minors of mat and find two additional points.

twocub=ideal submatrix(gens minors(2,mat),,{0,1})

quad=super basis(2,saturate(twocub,ideal I7)) ---four conics in x_i

R1=KK[x_0..x_2,y_0..y_3,g_0..g_8]

x1=matrix{{x_0..x_2}}

x3=symmetricPower(3,matrix{{x_0..x_2}})

y1=matrix{{y_0..y_3}}

---we find now p3, gale transform

p3=eliminate({x_0,x_1,x_2},sub(ideal I7,R1)+ideal(y1-sub(quad,R1)))

codim p3, degree p3

betti p3

----p3 is seven points in P^3, three quadrics and one cubic in y_i

----net3, only the three quadrics in y_i remain

net3=submatrix(gens p3,,{0..2})

codim ideal net3, degree ideal net3

i7net=eliminate({y_0,y_1,y_2,y_3},ideal(matrix{{x_0,x_1,x_2}}-net3)+

eliminate({y_3},p3))

codim i7net, degree i7net

dnet3=diff(symmetricPower(3,matrix{{x_0..x_2}}), transpose gens i7net)

gg=transpose genericMatrix(R1,g_0,3,3)

----we find now a 3*3 linear transformation from I7 to I7net

----we stack to I7, considered as 3*10 matrix, each of the cubics in I7net

--- (as a 10-dimensional vector), transformed with a unknown 3*3 matrix gg

---and we ask that the resulting 4*10 matrix is degenerate

-----if all the 4*4 minors are too many, one may select only some of them

IG=ideal(0_R1)

for j from 0 to 2 do IG=IG+minors(4,diff(x3,sub(sub((I7)_(0,j),R1),

apply(3,i->(x_i=>((gg)*transpose x1)_(i,0)))))||dnet3)

time SIG=saturate (IG,ideal(det(gg)));

codim SIG, degree SIG

----with Edge quartic we find three solutions,

--the following decompose command can be skipped if SIG vanishes on just 1 point

dgg=decompose SIG

gc=sub(matrix{apply(9,i->(g_i)%dgg_0)},g_8=>1)

gm=submatrix(gc,{0..2})||submatrix(gc,{3..5})||submatrix(gc,{6..8})

----gm is the projective transformation found

nq3=net3*(transpose gm)*transpose x1

----following mrep is the determinantal representation found

mrep=diff(transpose y1,diff( y1,nq3))

Algorithm 5

• INPUT: a quartic f

• OUTPUT: the image S(f) through the Scorza map

The script uses the expression of the Aronhold invariant in [17], example
1.2.1. Let H(fx) be the Hessian matrix of the cubic fx. All the 8-Pfaffians
of the following 9 × 9 matrix (written in 3 × 3 block form) coincide, up to
scalar, with the equation of S(f).

 0 H(fz) −H(fy)
−H(fz) 0 H(fx)
H(fy) −H(fx) 0


In the M2 script we check that starting from f = x4 + y4 + z4 + 2α(x3y+

y3z + z3x) − α(x2y2 + y2z2 + x2z2) + (α − 2
3
)(xy3 + yz3 + zx3) − 4(x2yz +

xy2z + xyz2), with α = −1+
√
−7

2
, then S(f) = xy3 + yz3 + zx3.

K2=toField ((ZZ/31991)[t]/ideal(t^2+7))

R=K2[y_0..y_2]

xx=basis(1,R)

a=(-1+t)/2

---change f in following line

f=y_0^4+y_1^4+y_2^4+2*a*(y_0^3*y_1+y_1^3*y_2+y_2^3*y_0)-a*(y_0^2*y_1^2+y_1^2*y_2^2+y_0^2*y_2^2)+

(a-2/3)*(y_0*y_1^3+y_1*y_2^3+y_2*y_0^3)-4*(y_0^2*y_1*y_2+y_0*y_1^2*y_2+y_0*y_1*y_2^2)

m=matrix{{0,y_2,-y_1},{-y_2,0,y_0},{y_1,-y_0,0}}

m9= diff(m,diff(xx,diff(transpose xx,f)));

scorzamap=(mingens pfaffians(8,m9))_(0,0)

---output is y_0*y_1^3+y_0^3*y_2+y_1*y_2^3

Algorithm 6

• INPUT: a quartic f and a point q ∈ S(f)

• OUTPUT: a determinantal representation corresponding to the image
(S(f), θ) through the Scorza map

The steps are the following

1. Compute the polar Pq(f), which is a Fermat curve since by assumption
q ∈ S(f). The Hessian determinant of Pq(f) was classically called the
Polihessian and it splits in three lines, its singular locus gives the three
points in S(f) corresponding to q in the (3, 3)-correspondence defined
by the θ. Add to the divisor given by these three points a generic
hyperplane divisor, call DP the resulting degree 7 divisor. We want to
find all the effective divisor of degree 6 linearly equivalent to DP−q. To
achieve this goal call SE = R/(S(f)) the quotient ring, call IP/(S(f))
and Iq/(S(f)) the images respectively of the ideals of DP and q in the
quotient ring , pick the first generator h of IP/(S(f)) and compute
h · Iq/(S(f)) : IP/(S(f)). This is an ideal generated by four cubics
that represent the four generators of H0(C,K + θ), in the following

sense. The common base locus of Ci is a degree six effective divisor on
S(f), say K + θ. Each cubic of the system vanishes on K + θ and in
additional six points which define a degree six effective divisor linearly
equivalent. Call Ci , for i = 0, . . . , 3, the representatives of these cubics
in the original ring R. Introduce new variables u0, . . . , u3 and eliminate
(x, y, z) from the ideal generated by S(f) and ui − Ci(x, y, z). We get
the ideal of the sextic curve in P3 which is defined by the linear system
K + θ.

2. The resolution of the ideal of the sextic curve in P3 found at previous
step gives a 4 × 3 C matrix with linear entries. Note that a different
flattening gives a 4× 4 matrix with linear entries, but we have to find
a symmetric representative of it.

3. The correct relation between coordinates ui and (x, y, z) is given by the

relation C ·

xy
z

 = 0 (see 4.3). After elimination of ui we find that

the polynomial uiuj are cubic polynomials in (x, y, z). We construct
the matrix md with entries uiuj which are cubic polynomials in x, y, z
Main problem is to correctly scale them To scale we find separately the
cubic polynomial corresponding to (ui + uj)

2 (recorded in matrix ad)
and we use the identity (ui + uj)

2 = u2i + u2j + 2uiuj

4. We conclude last step with a symmetric 4 × 4 matrix pmd with (i, j)
entry given by the cubic polynomial corresponding to uiuj. Now we
use Dixon technique, as explained in [28], we adjugate pmd and we
factor out (S(f))2 from each entry. The output is the wished linear
determinantal representation.

A nice example is to start from
f = x4 + y4 + z4 + (−6

√
2− 6) (x2y2 + x2z2 + y2z2)

which satisfies S(f) = x4 + y4 + z4.
Choosing the point q = (1, 0, e(π/4)

√
−1) ∈ S(f) we get the following

symmetric determinantal representation of the Fermat quartic (set α =√
−2 +

√
−1 and β =

√
−2−

√
−1)


(−α− 1)x + (β − 2

√
2 + 5)z (α + 1)y (β − 1)x + (−β − 1)z (−β + 1)y

(α + 1)y (β − 1)x + (2
√
−2−

√
−1−

√
2 + 3)z (−β + 1)y (α + 1)x + (−α + 3)z

(β − 1)x + (−β − 1)z (−β + 1)y (α + 1)x + (α− 1)z (−α− 1)y

(−β + 1)y (α + 1)x + (−α + 3)z (−α− 1)y (−β + 1)x + (
√
−1−

√
2 + 1)z

Actually the determinant is −32
√
−1 (x4 + y4 + z4).

A second example is the quartic f with Waring decomposition given by
f = x4 + y4 + (x+ y)4 + (x+ y + z)4 + (x+ 2y + 3z)4 + (−5x+ 7y − 11z)4.
Note that the point (0, 0, 1) belongs to the first three summands, so that the
polar cubic fz is Fermat and (0, 0, 1) ∈ S(f). The output found with the
algorithm is messy but can be found in less than one minute on a PC. We
do not print it here.

The M2 script is the following

------------------INPUT: f, and a point q in S(f)

------------------OUTPUT: a determinantal representation of S(f)

K2=QQ

--K2=toField(QQ[t]/ideal(t^2-2))

R=K2[x,y,z]

x2=basis(2,R)

x1=basis(1,R)

f=x^4+y^4+(x+y)^4+(x+y+z)^4+(x+2*y+3*z)^4+(-5*x+7*y-11*z)^4---the point q=(0,0,1)

--- kills first three summands, so it belongs to S(f)

-----

hes=det diff(transpose x1,diff(x1,diff(z,f)))---it is the polohessian triangle

singhes=ideal(diff(x1,hes))

---computation of S(f), called scf

m=matrix{{0,z,-y},{-z,0,x},{y,-x,0}}

scf= (mingens(pfaffians(8,diff(x1,diff(transpose x1,diff(m,f))))))_(0,0)

sub(scf,{x=>0,y=>0,z=>1})

----we have the theta given by singhes-PP

---now we add the hyperplane divisor and we find a degree six

---effective divisor linearly equivalent

SE=R/(scf)

DN= ideal(x,y)--negative part of divisor (1 point)

DP= intersect(sub(singhes,SE),ideal(random(1,SE)))--positive part

---of divisor (3+4=7 points)

h=(gens DP)_(0,0)

LD=((h*DN) : DP);---four cubics which have base points

---on the effective divisor and give the embedding of scf in P^3

betti LD

---from the four cubics vanishing on the six points on scf

---we get the sextic embedding in P^3

----by computing linear relation between (u,x) with Groebner basis

R2=K2[x,y,z,u_0..u_3]

ld=sub(LD,R2)

J=ideal(sub(scf,R2),u_0-(gens ld)_(0,0),u_1-(gens ld)_(0,1),

u_2-(gens ld)_(0,2),u_3-(gens ld)_(0,3))

----we offer two alternatives to compute the 4*3 cs2 matrix

---the first one, commented below is more direct

--rJ=res eliminate({x,y,z},J)

--cs2=(rJ.dd)_2;

---the second alternative turns out to be computationally cheaper

JJ=gens gb(J);

betti JJ

cs1=diff(transpose matrix{{x,y,z}},submatrix(JJ,,{0..3}));

cs2=transpose cs1;

-----cs2 is the 4*3 matrix

----the following commands reconstruct the 4*4 determinantal representation

----from the 4*3 matrix, which should be "symmetrizable"

------we want first to construct the matrix pmd with entries u_i*u_j

--- which are cubic polynomials in x,y,z

------main problem is to correctly scale them

------to scale we use the identity (u_i+u_j)^2=u_i^2+u_j^2+2*u_i*u_j

----md is the matrix pmd, up to scale, this elimination procedure is crucial

--- and works only for "symmetrizable" 4*3 matrix

md=mutableMatrix(R2,4,4)

--next loop runs in some seconds

for i from 0 to 3 do for j from i to 3 do

md_(i,j)=(gens eliminate({u_0,u_1,u_2,u_3},saturate(ideal(cs2*

transpose matrix{{x,y,z}} )+ideal(u_i*u_j),ideal(u_0,u_1,u_2,u_3))))_(0,0)

for i from 0 to 3 do for j from 0 to i-1 do md_(i,j)=md_(j,i)

-----ad contains the auxiliary polynomials (u_i+u_j)^2

ad=mutableMatrix(R2,4,4)

for i from 0 to 3 do for j from i to 3 do

ad_(i,j)=(gens eliminate({u_0,u_1,u_2,u_3},saturate(ideal(

cs2*transpose matrix{{x,y,z}} )+ideal((u_i+u_j)^2),

ideal(u_0,u_1,u_2,u_3))))_(0,0)

-----S2 is the ring with the coefficients p_(i,j), q_(i,j) needed to scale

S2=K2[x,y,z,u_0..u_3,p_(0,0)..p_(3,3),q_(0,0)..q_(3,3)]

md=sub(matrix md,S2)

ad=sub(matrix ad,S2)

x3=symmetricPower(3,matrix{{x,y,z}})

-----I contains the conditions needed in order to scale

I=ideal()

for i from 0 to 2 do for j from i+1 to 3 do

I=I+ideal(diff(x3,p_(i,i)*md_(i,i)+p_(j,j)*md_(j,j)+

2*p_(i,j)*md_(i,j)-q_(i,j)*ad_(i,j)))

codim I, degree I

------pp contains the solution to the system given by I,

---this works if q_(2,3) is different from zero,

---otherwise the solution to I has to be extracted in alternative way

pp=mutableMatrix(S2,4,4)

for i from 0 to 3 do for j from i to 3 do pp_(i,j)=(sub(p_(i,j)%I,q_(2,3)=>1))

pp

------pmd is our goal matrix

pmd=mutableMatrix(S2,4,4)

for i from 0 to 3 do for j from i to 3 do

pmd_(i,j)=pp_(i,j)*md_(i,j)

for i from 0 to 3 do for j from 0 to i-1 do pmd_(i,j)=pmd_(j,i)

pmd=matrix(pmd)

------now we use Dixon technique, we adjugate pmd

---and we factor out (scf)^2 from each entry

mdf=mutableMatrix(S2,4,4)

for i from 0 to 3 do for j from 0 to 3 do mdf_(i,j)=(-1)^(i+j)*

(quotientRemainder(det submatrix’(pmd,{i},{j}),sub((scf)^2,S2)))_0

---mdf is our output

mdf=matrix mdf

factor det mdf

Algorithm 7

• INPUT: a determinantal representation of a quartic g corresponding to
(g, θ)

• OUTPUT: the quartic f such that the image (S(f), θ) through the
Scorza map corresponds to (g, θ)

The steps are the following

1. We find the sextic model of the quartic in P3 by transformimg the
4× 4× 3 representation in a 4× 3 matrix M(u) with linear entries in
u = (u0, . . . , u3). The maximal minors of M define the sextic model
given by the linear system K + θ.

2. We define the (3, 3) correspondence introducing dual coordinates v =
(v0, . . . , v3) and the ideal cor33 generated by the maximal minors of
M(u), M(v) and by the bilinear equations given by u·M(v) and v·M(u).

3. We lift the (3, 3)-correspondence to the quartic model with dual coor-
dinates a = (a0, . . . , a2) and b = (b0, . . . , b2). This works by eliminating
u and v from the saturation over u and v of the ideal generated by
cor33 and by the bilinear equations M(u) · at and M(v) · bt. We get
6 biquadratic equations in (a, b) that we call eli6 This step is com-
putationally expensive. At the end we convert the six equations in a
6× 36 matrix, where the 36 columns correspond to the 36 biquadratic
monomials in (a, b), called maab.

4. Let fc be a generic quartic given by 15 entries ci, we compute the 6
biquadratic equations in (a, b) obtained by the condition rk(PaPbfc) ≤
1. We convert these equations in a 6 × 36 matrix, called mab, with
entries depending quadratically on ci.

5. We impose that the rows of two matrices maab and mab span the same
6-dimensional space. This gives quadratical equations in ci. Note that
Scorza map has entries which are quartic in ci, so that this “tour de
force” has an effective gain at the end. In principle this can be obtained
by stacking each row of mab to maab and computing all the (maximal)
7×7-minors. This is computationally too expensive, so we choose only
some of these minors. For general quartics, the minors which contain
the first 6 columns suffice. In the M2 script it is sketched an alternative
choice of minors which works for the Klein quartic.

6. The ideal of the quadratical equations in ci obtained in previous step
has to be saturated with base locus of Scorza map. The output is a
unique point in ci which gives our output quartic. If the output is
bigger, we have to repeat the previous step by adding more minors.

As a running example, we may start from

A =


0 x− 2y z − 2x 2z + y

x− 2y 0 −2z + y z + 2x
z − 2x −2z + y 0 x+ 2y
2z + y z + 2x x+ 2y 0

 (9.1)

with detA = 25(x4 + y4 + z4)− 34(x2y2 + x2z2 + y2z2) and find as output

x4 + y4 + z4 + 30(x2y2 + x2z2 + y2z2). (9.2)

The algorithm can be iterated for other 35 times, as described in [28]. More
precisely we first find the bitangent matrix of (9.1) with Algorithm 3, then
we apply [28, Theorem 3.9] with all {0, i, j, k} such that {i, j, k} ⊂ {1..7}.
All 36 quartics are real, in agreement with (1.3).

1. mother case, x4 + 30x2y2 + y4 + 30x2z2 + 30y2z2 + z4

2. (0, 1, 2, 3), x4 + (10/3)x2y2 + y4 + (10/3)x2z2 + (10/3)y2z2 + z4)

3. (0, 1, 2, 4), x4−(84/125)x3y+(54/25)x2y2+(492/125)xy3+(81/625)y4+
(28/25)x3z − (36/5)x2yz − (108/25)xy2z − (492/125)y3z + 6x2z2 +
(36/5)xyz2 + (54/25)y2z2 + (28/25)xz3 + (84/125)yz3 + z4)

4. (0, 1, 2, 5), (625/81)x4 + (700/81)x3y + (1250/27)x2y2 + (700/81)xy3 +
(625/81)y4 +(140/27)x3z+(500/9)x2yz−(500/9)xy2z−(140/27)y3z+
(50/3)x2z2−(100/3)xyz2+(50/3)y2z2−(820/27)xz3+(820/27)yz3+z4)

5. (0, 1, 2, 6), x4 + (164/9)x3y + 6x2y2 + (164/9)xy3 + y4 + (164/9)x3z +
12x2yz+ 12xy2z+ (164/9)y3z+ 6x2z2 + 12xyz2 + 6y2z2 + (164/9)xz3 +
(164/9)yz3 + z4)

6. (0, 1, 2, 7), (81/625)x4−(492/125)x3y+(54/25)x2y2+(84/125)xy3+y4+
(492/125)x3z−(108/25)x2yz+(36/5)xy2z+(28/25)y3z+(54/25)x2z2−
(36/5)xyz2 + 6y2z2 − (84/125)xz3 + (28/25)yz3 + z4)

7. (0, 1, 3, 4), (81/625)x4−(492/125)x3y+(54/25)x2y2+(84/125)xy3+y4−
(492/125)x3z+(108/25)x2yz−(36/5)xy2z−(28/25)y3z+(54/25)x2z2−
(36/5)xyz2 + 6y2z2 + (84/125)xz3 − (28/25)yz3 + z4)

8. (0, 1, 3, 5), x4 + (164/9)x3y + 6x2y2 + (164/9)xy3 + y4 − (164/9)x3z −
12x2yz− 12xy2z− (164/9)y3z+ 6x2z2 + 12xyz2 + 6y2z2− (164/9)xz3−
(164/9)yz3 + z4)

9. (0, 1, 3, 6), (625/81)x4 + (700/81)x3y + (1250/27)x2y2 + (700/81)xy3 +
(625/81)y4−(140/27)x3z−(500/9)x2yz+(500/9)xy2z+(140/27)y3z+
(50/3)x2z2−(100/3)xyz2+(50/3)y2z2+(820/27)xz3−(820/27)yz3+z4)

10. (0, 1, 3, 7), x4−(84/125)x3y+(54/25)x2y2+(492/125)xy3+(81/625)y4−
(28/25)x3z+(36/5)x2yz+(108/25)xy2z+(492/125)y3z+6x2z2+(36/5)xyz2+
(54/25)y2z2 − (28/25)xz3 − (84/125)yz3 + z4)

11. (0, 1, 4, 5), x4−(15/16)y4+(17/4)x3z+(45/4)xy2z+(3/2)x2z2+(17/4)xz3+
z4)

12. (0, 1, 4, 6), 15x4 + y4 + 45x2yz − 17y3z + 24y2z2 − 17yz3 + z4)

13. (0, 1, 4, 7), (9/26)x4+(15/13)x2y2+(9/26)y4−(15/13)x2z2−(15/13)y2z2+
z4)

14. (0, 1, 5, 6),−(1/16)x4−(15/8)x2y2−(1/16)y4+(15/8)x2z2+(15/8)y2z2+
z4)

15. (0, 1, 5, 7), 15x4 + y4 − 45x2yz + 17y3z + 24y2z2 + 17yz3 + z4)

16. (0, 1, 6, 7), x4−(15/16)y4−(17/4)x3z−(45/4)xy2z+(3/2)x2z2−(17/4)xz3+
z4)

17. (0, 2, 3, 4), x4 − (164/9)x3y + 6x2y2 − (164/9)xy3 + y4 + (164/9)x3z −
12x2yz+ 12xy2z− (164/9)y3z+ 6x2z2− 12xyz2 + 6y2z2 + (164/9)xz3−
(164/9)yz3 + z4)

18. (0, 2, 3, 5), (81/625)x4+(492/125)x3y+(54/25)x2y2−(84/125)xy3+y4+
(492/125)x3z+(108/25)x2yz+(36/5)xy2z−(28/25)y3z+(54/25)x2z2+
(36/5)xyz2 + 6y2z2 − (84/125)xz3 − (28/25)yz3 + z4)

19. (0, 2, 3, 6), x4+(84/125)x3y+(54/25)x2y2−(492/125)xy3+(81/625)y4+
(28/25)x3z+(36/5)x2yz−(108/25)xy2z+(492/125)y3z+6x2z2−(36/5)xyz2+
(54/25)y2z2 + (28/25)xz3 − (84/125)yz3 + z4)

20. (0, 2, 3, 7), (625/81)x4 − (700/81)x3y + (1250/27)x2y2 − (700/81)xy3 +
(625/81)y4 +(140/27)x3z−(500/9)x2yz−(500/9)xy2z+(140/27)y3z+
(50/3)x2z2+(100/3)xyz2+(50/3)y2z2−(820/27)xz3−(820/27)yz3+z4)

21. (0, 2, 4, 5), (1/15)x4 +(17/15)x3y+(8/5)x2y2 +(17/15)xy3 +(1/15)y4−
3xyz2 + z4)

22. (0, 2, 4, 6), x4 − 30x2y2 − 16y4 + 30x2z2 − 30y2z2 + z4)

23. (0, 2, 4, 7),−(15/16)x4 + y4 + (45/4)x2yz + (17/4)y3z + (3/2)y2z2 +
(17/4)yz3 + z4)

24. (0, 2, 5, 6),−(15/16)x4 + y4 − (45/4)x2yz − (17/4)y3z + (3/2)y2z2 −
(17/4)yz3 + z4)

25. (0, 2, 5, 7), x4 − (10/3)x2y2 + (26/9)y4 + (10/3)x2z2 − (10/3)y2z2 + z4)

26. (0, 2, 6, 7), (1/15)x4−(17/15)x3y+(8/5)x2y2−(17/15)xy3 +(1/15)y4 +
3xyz2 + z4)

27. (0, 3, 4, 5),−16x4 − 30x2y2 + y4 − 30x2z2 + 30y2z2 + z4)

28. (0, 3, 4, 6),−(16/15)x4−(68/15)x3y−(8/5)x2y2−(68/15)xy3−(16/15)y4−
12xyz2 + z4)

29. (0, 3, 4, 7), x4 + 15y4 − 17x3z + 45xy2z + 24x2z2 − 17xz3 + z4)

30. (0, 3, 5, 6), x4 + 15y4 + 17x3z − 45xy2z + 24x2z2 + 17xz3 + z4)

31. (0, 3, 5, 7),−(16/15)x4+(68/15)x3y−(8/5)x2y2+(68/15)xy3−(16/15)y4+
12xyz2 + z4)

32. (0, 3, 6, 7), (26/9)x4 − (10/3)x2y2 + y4 − (10/3)x2z2 + (10/3)y2z2 + z4)

33. (0, 4, 5, 6), x4 − (164/9)x3y + 6x2y2 − (164/9)xy3 + y4 − (164/9)x3z +
12x2yz− 12xy2z+ (164/9)y3z+ 6x2z2− 12xyz2 + 6y2z2− (164/9)xz3 +
(164/9)yz3 + z4)

34. (0, 4, 5, 7), (81/625)x4+(492/125)x3y+(54/25)x2y2−(84/125)xy3+y4−
(492/125)x3z−(108/25)x2yz−(36/5)xy2z+(28/25)y3z+(54/25)x2z2+
(36/5)xyz2 + 6y2z2 + (84/125)xz3 + (28/25)yz3 + z4)

35. (0, 4, 6, 7), x4+(84/125)x3y+(54/25)x2y2−(492/125)xy3+(81/625)y4−
(28/25)x3z − (36/5)x2yz + (108/25)xy2z − (492/125)y3z + 6x2z2 −
(36/5)xyz2 + (54/25)y2z2 − (28/25)xz3 + (84/125)yz3 + z4)

36. (0, 5, 6, 7), (625/81)x4 − (700/81)x3y + (1250/27)x2y2 − (700/81)xy3 +
(625/81)y4−(140/27)x3z+(500/9)x2yz+(500/9)xy2z−(140/27)y3z+
(50/3)x2z2+(100/3)xyz2+(50/3)y2z2+(820/27)xz3+(820/27)yz3+z4)

The automorphism group of detA = 25(x4 + y4 + z4)− 34(x2y2 + x2z2 +
y2z2) is the octahedral group S4 of order 24. The 36 preimages can be
grouped into the S4-orbits which are described in the following table.

(0, i, j, k) topological classification Automorphism group

mother case in (9.2) empty S4 of order 24

(0, 1, 2, 3) empty S4 of order 24

(0, 1, 2, 4)(0, 1, 3, 7)(0, 2, 3, 6)(0, 4, 6, 7)
(0, 1, 2, 5)(0, 1, 3, 6)(0, 2, 3, 7)(0, 5, 6, 7)
(0, 1, 2, 7)(0, 1, 3, 4)(0, 2, 3, 5)(0, 4, 5, 7)

one oval ZZ2 of order 2

(0, 1, 4, 5)(0, 1, 6, 7)(0, 2, 4, 7)
(0, 2, 5, 6)(0, 3, 4, 6)(0, 3, 5, 7)

one oval ZZ2 ⊕ ZZ2 of order 4

(0, 1, 4, 6)(0, 1, 5, 7)(0, 2, 4, 5)
(0, 2, 6, 7)(0, 3, 4, 7)(0, 3, 5, 6)

one oval ZZ2 ⊕ ZZ2 of order 4

(0, 1, 2, 6)(0, 1, 3, 5)(0, 2, 3, 4)(0, 4, 5, 6) one oval S3 of order 6

(0, 1, 4, 7)(0, 2, 5, 7)(0, 3, 6, 7) empty D4 of order 8

(0, 1, 5, 6)(0, 2, 4, 6)(0, 3, 4, 5) one oval D4 of order 8

Figure 1: One of the 36 Scorza preimages of the Edge quartic

We print the figure corresponding to the star corresponding to (0, 1, 5, 6),
provided by Emanuele Ventura.

In the case of the Klein quartic we can follow an alternative path. Note
that the 28 bitangents of the Klein quartic are printed in [33]. Only four of
them are real. The Klein quartic f has the feature that

f = S(f). (9.3)

Ciani proves in [7] that the only quartics f satisfying (9.3) are the Klein
quartic and the double conics (see also [8])1. The other 35 preimages of
Klein quartic through the Scorza map (besides the Klein quartic itself) were
found by Ciani in [6], although Ciani describes them intrinsically and he does
not provide the equations. They divide in three SL(3)-classes, two classes
(conjugate to each other) containing seven quartics each one, having Aut
the octahedral group of order 24, and a (self-conjugate) class containing 21
quartics, having Aut = D4 the dihedral group of order 8. This class contains
three real quartics by (1.3). We have found the following solution (among the

1It should be interesting to compute the weight decomposition of the tangent space at
the fixed points of S, in order to understand the dynamics of the iteration of the Scorza
map.

possible 35), having octahedral symmetry. Let α = τ + τ 2 + τ 4 = −1+
√
−7

2
. If

f = x4 +y4 +z4 +2α(x3y+y3z+z3x)−α(x2y2 +y2z2 +x2z2)+(α− 2
3
)(xy3 +

yz3 + zx3)− 4(x2yz+ xy2z+ xyz2) then S(f) = xy3 + yz3 + zx3 is the Klein
quartic. The Klein quartic belongs to the pencil generated by the double
(invariant) conic [x2 + y2 + z2 + α(xy + xz + yz)]

2
and by the quadrilateral

(given by four bitangents) (x + y + z)(x + (τ 5 + τ 4 + 2τ)y + (−2τ 5 − τ 4 +
τ 3 − τ 2 − 2 ∗ τ)z)((−2τ 5 − τ 4 + τ 3 − τ 2 − 2τ)x+ y + (τ 5 + τ 4 + 2τ)z)((τ 5 +
τ 4 + 2τ)x+ (−2τ 5 − τ 4 + τ 3 − τ 2 − 2τ)y + z).

The M2 script is the following

KK=QQ

R=KK[x,y,z,u_0..u_3]

A = matrix({{0, x + 2*y, z + 2*x, -2*z + y},

{x + 2*y, 0, 2*z + y, z - 2*x},

{z + 2*x, 2*z + y, 0, x - 2*y},

{-2*z + y, z - 2*x, x - 2*y, 0}});

----output will be x^4+30*x^2*y^2+y^4+30*x^2*z^2+30*y^2*z^2+z^4

det(A)

----modification M’_{1234}, according to [Hesse]

A=matrix{{0,x-2*y,z-2*x,2*z+y},

{x-2*y,0,-2*z+y,z+2*x},

{z-2*x,-2*z+y,0,x+2*y},

{2*z+y,z+2*x,x+2*y,0}}

-----output will be

-- x^4+(10/3)*x^2*y^2+y^4+(10/3)*x^2*z^2+(10/3)*y^2*z^2+z^4

uu=transpose matrix {{u_0..u_3}}

--following M gives the sextic model in P^3 with coordinates u

M=diff(x,A)*uu|diff(y,A)*uu|diff(z,A)*uu

R2=KK[a_0..a_2,b_0..b_2,u_0..u_3,v_0..v_3,MonomialOrder=>{3,3,4,4}]

aa=matrix{{a_0..a_2}}, bb=matrix{{b_0..b_2}}

cs2=sub(M,R2)

----following gives the (3,3) correspondence on the sextic model with coord.(u,v)

cor33=minors(3,cs2)+sub(minors(3,cs2),apply(4,i->(u_i=>v_i)))+

ideal(matrix{{v_0..v_3}}*cs2)+

ideal(matrix{{u_0..u_3}}*sub(cs2,apply(4,i->(u_i=>v_i))))

----following lifts the (3,3) corresp. to the quartic model with coord. (a,b)

----it is a long computation, running in about 15 minutes on a PC,

time eli=eliminate({u_0,u_1,u_2,u_3,v_0,v_1,v_2,v_3},

saturate(ideal(cs2*transpose aa)+ideal(sub(cs2,apply(4,i->(u_i=>v_i)))*

transpose bb)+cor33,ideal(u_0..u_3)*ideal(v_0..v_3)))

eli6=submatrix(gens eli,,{1..6})

-----the (3,3) correspondence is given by 6 biquadratic equations in a,b in eli6

R4=KK[x,y,z,a_0..a_2,b_0..b_2,c_0..c_14]

---now we compare the correspondence found with the correspondence obtained

---starting from a generic quartic fc

---given by rank (P_aP_b(fc)) <= 1

fc=(matrix{{c_0..c_14}}*transpose symmetricPower(4,matrix{{x,y,z}}))_(0,0)

f1=a_0*diff(x,fc)+a_1*diff(y,fc)+a_2*diff(z,fc)

f2=b_0*diff(x,f1)+b_1*diff(y,f1)+b_2*diff(z,f1)

xx=matrix{{x,y,z}}

ab=minors(2,diff(transpose xx,diff(xx,f2)));

---previous ab is the correspondence obtained from fc

----next aab is our correspondence

aab=ideal sub(eli6,R4);

----now to compare the two correspondences we encode in matrices with 36 columns

---given by all biquadratic monomials

ab2=symmetricPower(2,matrix{{a_0..a_2}})**symmetricPower(2,matrix{{b_0..b_2}})

maab=diff(transpose ab2, gens aab)

---following must be nonzero

det submatrix(maab,{0..5},)

det submatrix(maab,{0,1,4,7,10,14},)

mab=diff(transpose ab2, gens ab);

----we impose that each column from mab is in the column span of maab

----in principle we have to check all 7*7 minors, but this is too expensive

----so we try with all 7*7 minors containing the first six rows, this is enough

---in the examples found (not for Klein quartic!)

--if at the end we get infinitely many solutions, one has to add more conditions

---given by more minors

-----------------following in case of Klein quartic

I=ideal(apply(36,i->det submatrix(maab|mab_{0},{0,1,4,7,10,14,i},)));

I=I+ideal(apply(36,i->det submatrix(maab|mab_{1},{0,1,4,7,10,14,i},)));

I=I+ideal(apply(36,i->det submatrix(maab|mab_{2},{0,1,4,7,10,14,i},)));

I=I+ideal(apply(36,i->det submatrix(maab|mab_{3},{0,1,4,7,10,14,i},)));

I=I+ideal(apply(36,i->det submatrix(maab|mab_{4},{0,1,4,7,10,14,i},)));

I=I+ideal(apply(36,i->det submatrix(maab|mab_{5},{0,1,4,7,10,14,i},)));

I=I+ideal(apply(36,i->det submatrix(maab|mab_{6},{0,1,4,7,10,14,i},)));

I=I+ideal(apply(36,i->det submatrix(maab|mab_{7},{0,1,4,7,10,14,i},)));

I=I+ideal(apply(36,i->det submatrix(maab|mab_{8},{0,1,4,7,10,14,i},)));

betti I

----------------------until here for Klein quartic

I=ideal(apply(30,i->det submatrix(maab|mab_{0},{0,1,2,3,4,5,i+6},)));

I=I+ideal(apply(30,i->det submatrix(maab|mab_{1},{0,1,2,3,4,5,i+6},)));

I=I+ideal(apply(30,i->det submatrix(maab|mab_{2},{0,1,2,3,4,5,i+6},)));

I=I+ideal(apply(30,i->det submatrix(maab|mab_{3},{0,1,2,3,4,5,i+6},)));

I=I+ideal(apply(30,i->det submatrix(maab|mab_{4},{0,1,2,3,4,5,i+6},)));

I=I+ideal(apply(30,i->det submatrix(maab|mab_{5},{0,1,2,3,4,5,i+6},)));

I=I+ideal(apply(30,i->det submatrix(maab|mab_{6},{0,1,2,3,4,5,i+6},)));

I=I+ideal(apply(30,i->det submatrix(maab|mab_{7},{0,1,2,3,4,5,i+6},)));

I=I+ideal(apply(30,i->det submatrix(maab|mab_{8},{0,1,2,3,4,5,i+6},)));

betti I

sI=saturate I;

betti sI

----sI contains now the base locus of Scorza map plus one single point,

----we have to saturate with base locus of Scorza, called bscorza

m=matrix{{0,z,-y},{-z,0,x},{y,-x,0}}

m9= diff(m,diff(xx,diff(transpose xx,fc)));

scorza=(mingens pfaffians(8,m9))_(0,0);

bscorza=ideal diff(symmetricPower(4,xx),scorza);

fiber=saturate(sI,bscorza)

----the following gives the quartic in the output

(sub(matrix {apply(15,i->(c_i)%fiber)},c_14=>1)*transpose symmetricPower(4,matrix{{x,y,z}}))_(0,0)

Algorithm 8

• INPUT: a plane quartic f

• OUTPUT: the order of the automorphism group of linear invertible
transformations which leave f invariant

This algorithm is straightforward. We add here the M2 script for the
convenience of the reader. The structure of groups of small size can be found
by counting the elements of a given order.

---work on a finite field to run quicker

R2=ZZ/31991[g_0..g_8,y_0..y_2]

gg=transpose genericMatrix(R2,3,3)

-----we impose conditions to a unknown projective transformation gg

y4=symmetricPower(4,matrix{{y_0..y_2}})

y1=transpose matrix{{y_0..y_2}}

f=y_0^4+30*y_0^2*y_1^2+y_1^4+30*y_0^2*y_2^2+30*y_1^2*y_2^2+y_2^4 ---order 24

IG=minors(2,diff(y4,sub(sub(f,R2),apply(3,i->(y_i=>((gg)*y1)_(i,0)))))||

diff(y4,f));

time sig=saturate(IG,ideal(det(gg)));

codim sig, degree sig

---if codim sig=8, then degree sig is the order of automorphism group

Università di Firenze, viale Morgagni 67/A, 50134 Firenze (Italy). e-mail:
ottavian@math.unifi.it

45


